Adding Printers via Logon Scripts

Up until recently, my organizations network printers were scattered to the four winds and controlled by miscellaneous workstations around the office. This meant if any one workstation should fail or go offline, access to that network printer went with it. I began planning to deploy a printer server but wanted to deploy the connections to the printer without having to visit every workstation.

I was very interested to see Bill Stewarts article on per machine printer adding in the December issue of Windows IT Pro. I have been looking for a more elegant printer deployment solution for workstations and had a very specific set of goals.

1. To assign printers to computers irregardless of whom the logged on user was.
2. Remove irrelevant or old printer mappings without removing the local / hardwired printers.
3. Assign default printer based on user specification and the available connected printers.
4. The assignment of the printers to the computer happens without doing anything more than joining a computer to a domain and dropping it in it's correct Organizational Unit the in Active Directory

As I delved into the article it became clear to me that this solution, while elegant, would not suite my organizations needs. In recent weeks, my organizations logon scripts had become more and more complex, assigning network resources such as mapped drives based on security groups. I had also recently cleaned up and organized the active directory, placing computers and users into organizational units. In the past, I have assigned printers to a computer, but it was nothing more than a simple:

```
objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE"
```

This was a non-qualified printer addition and happened for everybody across the enterprise, no matter what OU they were in. My company had offices in several locations and I have broken down the Active Directory to reflect that. Now the question was how could I assign a printer resource based on the OU the computer was in?

I knew that I could get the OU a user belonged to, but if I could get the OU the computer itself was in I could start to assign computer based resources, such as printers.

The ADSystemInfo object was the key. Using it you can get the full AD path of the computer, which contains the OU name you could be looking for.

Script Example 1:

```
Set objADInfo  = CreateObject("ADSystemInfo")
strUserOU      = objADInfo.UserName
strComputerOU  = objADInfo.ComputerName
```

Now armed with this information, you can begin assign resources based on use by using the InStr command to determine if the OU name you are looking for is in the string strComputerOU.

Script Example 2:

```
If InStr(1, strComputerOU,"OU=Target OU") > 0 Then
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE1"
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE2"
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE3"
End If
```

Now this starts to get you where you want to be. But there are many miscellaneous rouge network printer connections on the workstations. Perhaps they were set using the IP address or the server name. The fact is that these need to be removed from the end users workstations.

At first I considered doing some sort of registry hack to accomplish this but was leery of launching any sort of registry modification from a logon script or having end users run a registry update themselves. I then found using the wscript.network object that I could enumerate through all the printers on a local computer. How ever this returned all printers, including local printers to that computer. What I wanted to do was only remove network printer connections so I had to do a little filtering to find the “\\” at the beginning of a network printer resource.

Script Example 3:

```
Set objNetwork  = CreateObject("Wscript.Network")
Set objADInfo   = CreateObject("ADSystemInfo")
strUserOU       = objADInfo.UserName
strComputerOU   = objADInfo.ComputerName
Set objPrinters = objNetwork.EnumPrinterConnections

If InStr(1, strComputerOU,"OU=Target OU") > 0 Then
  For Counter = 0 to objPrinters.Count - 1
    If mid(objPrinters.Item(Counter + 1), 1, 2) = "\\" Then
      PrinterPath = objPrinters.Item(Counter + 1)
      objNetwork.RemovePrinterConnection PrinterPath, True, True
    End If
  Next
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE1"
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE2"
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE3"
```

```
End If
```

Now the printer assignment portion of the logon script was starting to take shape. There was just one task left to complete. Each user needed to be assigned a default network printer. While we have a standard printer in the office that everyone should print to, many requested a different default printer. (The printer that the print job is immediately sent to after clicking the print icon and not going through “File” > “Print”)

I knew I could do this but based on user preferences seemed exceptionally difficult. Then I realized, why not created default printer Groups and add members to those groups. For example, I created a “Default HP 4100” users group for those who wanted to print by default to the networked HP 4100 printer, a group called “Default Xerox 6120” for those who wanted to print to it by default and so on.

I then quickly added users to the appropriate group and then got back to modifying the script. I added code to go through all the groups for a user, if they belonged to the appropriate default printer group, then I assigned them the correct default printer.

Script Example 4:

```
Set objNetwork  = CreateObject("Wscript.Network")
Set objADInfo   = CreateObject("ADSystemInfo")
strUserOU       = objADInfo.UserName
strComputerOU   = objADInfo.ComputerName
Set objPrinters = objNetwork.EnumPrinterConnections
strUser         = objNetwork.UserName
strDomain       = “DOMAIN”

If InStr(1, strComputerOU,"OU=Target OU") > 0 Then
  For Counter = 0 to objPrinters.Count - 1
    If mid(objPrinters.Item(Counter + 1), 1, 2) = "\\" Then
      PrinterPath = objPrinters.Item(Counter + 1)
      objNetwork.RemovePrinterConnection PrinterPath, True, True
    End If
  Next
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE1"
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE2"
  objNetwork.AddWindowsPrinterConnection "\\SERVER\PRINTERSHARE3"
  Set objGroups = GetObject("WinNT://"& strDomain & "/" & strUser & ",user")
    For Each objGroups In objGroups.Groups
      If objGroups.Name = "Default Printer 1 Users" Then
        objNetwork.SetDefaultPrinter "\\SERVER\PRINTERSHARE1"
      ElseIf objGroups.Name = "Default Printer 2 Users" Then
        objNetwork.SetDefaultPrinter "\\SERVER\PRINTERSHARE2"
      ElseIf objGroups.Name = "Default Printer 3 Users" Then
        objNetwork.SetDefaultPrinter "\\SERVER\PRINTERSHARE3"
      End If
    Next
End If
```

Now when a user logs on, the script removes old network printer mappings, assigns printers based on the OU the computer belongs and default by the group the user belongs too.

Now all I have to do when a new employee starts is simply create the user account and add them to the correct default printer group. When they logon, all the resources they need are there.